

~~TH-B132~~

TH-BD3

Stage DEUG INFO

Programme Saisie Portable

Type HUSKY

Mesure Sonde LPC - INRA

Stéphane MARMEY

Juin 1985

Je remercie, très réellement, Monsieur Patrick BERTUZZI du souci constant porté à la parfaite qualité de mon stage.

Son enseignement et la confiance témoignée m'ont permis d'acquérir connaissances et expérience pratique.

Il m'a efficacement aidé par ses conseils, et en y participant, à l'élaboration de ce rapport.

SOMMAIRE

L'I.N.R.A.	
L'I.N.R.A.	p. 2
Le Centre de Recherches d'Avignon	p. 2
La Station Science du Sol	p. 2
LA SONDE LPC-INRA	
Principe de la méthode	p. 3
POURQUOI L'UTILISATION D'UN PORTABLE ?	p. 5
PRESENTATION DU HUSKY-HUNTER	
Présentation	p. 6
Système	p. 7
DEMOS-Système	p. 7
Structure des fichiers et organisation mémoire	p. 7
Basic	p. 9
LE PROGRAMME	
Caractéristiques	p. 9
Présentation globale	p. 9
Présentation détaillée	p. 10
Entrée des mesures	p. 10
Edition des mesures	p. 13
Transfert des mesures	p. 14
CONCLUSION	p. 15
LISTE DES VARIABLES UTILISEES DANS LE PROGRAMME	p. 16
ORGANIGRAMMES	p. 18
PROGRAMMATION	p. 31

L'I.N.R.A.

Créé en 1946, l'I.N.R.A. (Institut National de Recherches Agronomiques) a reçu pour mission l'organisation, l'exécution et la publication de tous travaux de recherches scientifiques.

Pour cela, l'I.N.R.A. emploie plus de 8.000 personnes dont 2.000 scientifiques et ingénieurs regroupés dans 21 centres de recherches.

LE CENTRE DE RECHERCHES D'AVIGNON

Il fut créé en 1953 sur le domaine St Paul, puis étendu principalement sur le domaine St Maurice et les Vignères.

Situé au coeur d'une région d'intense production agricole, il comprend 20 services expérimentaux.

LA STATION SCIENCE DU SOL

Elle étudie la porosité du sol, son origine, sa morphologie, son évolution sous l'action de l'eau et des contraintes mécaniques, son influence sur les modalités d'irrigation fertilisante localisée.

LA SONDE LPC-INRA

Du point de vue de ses propriétés physiques, hydriques et mécaniques, le sol est un milieu poreux. L'espace poral est constitué par l'ensemble des pores ou des fissures. Celui-ci assure des fonctions fondamentales d'aération du sol et de transfert de l'eau et des sels dissous.

La mesure de l'espace poral ou porosité du sol est une mesure fondamentale de la Physique du Sol. Dans la pratique, cette mesure est le plus souvent indirecte. On mesure généralement la masse volumique sèche du sol considéré. C'est le rapport entre :

$$\varphi_d = \frac{\text{masse du sol sec}}{\text{volume du sol humide}}$$

Cette mesure est souvent réalisée sur le terrain par l'utilisation d'une sonde à transmission gamma.

La double sonde LPC-INRA est un gammadensimètre conçu et construit par le Centre d'Etudes et de Construction de Prototypes de l'Équipement d'Angers.

La contribution de l'I.N.R.A. et plus particulièrement de la Station Science du Sol d'Avignon dans son élaboration a constitué à établir un cahier des charges visant à adapter la méthode gammadensimétrique aux problèmes agronomiques et aux conditions pratiques de sa mise en oeuvre sur le terrain.

Principe de la méthode. (fig. 1)

La méthode de mesure gammadensimétrique de mesure de la masse volumique d'un sol en place est basée sur la loi d'atténuation par la matière d'un rayonnement gamma émis par une source radioactive.

Placé entre deux tubes, le sol est traversé par un rayonnement gamma émis par une source radioactive (A). Un dispositif de comptage (B) enregistre les photons non absorbés par le matériau. Plus le sol est poreux et sec, plus la mesure de comptage est faible. La densité humide est calculée grâce à une relation linéaire (comptage - densité) obtenue

lors de l'étalonnage de la sonde. Cette technique de mesure est particulièrement adaptée à l'étude des variations de la masse volumique sèche du sol, avec profondeur, dans les horizons superficiels du sol.




Fig.1 SONDE GAMMA TYPE LPC_INRA

POURQUOI L'UTILISATION D'UN PORTABLE ?

Un programme, appelé LPC.BAS, a été fait sur un micro-ordinateur portable Husky-Hunter, parfaitement adapté dans sa conception à la saisie des mesures obtenues à partir de la sonde sur le terrain.

Ce programme a été réalisé pour plusieurs raisons.

Un programme de traitement des mesures "sonde LPC-INRA" est implanté sur l'ordinateur central du Centre d'Avignon (MINI 6 - CII HONEYWELL BULL). Jusqu'ici, les données de base exploitées par le programme sont entrées au clavier d'une console MINI 6 : opération coûteuse en temps de saisie, de vérification de données rentrées.

L'utilisation d'une saisie portable, type Husky-Hunter, répond donc à deux objectifs :

- saisie directe des données sur le terrain,
- transfert automatique de celles-ci vers l'ordinateur de traitement.

Elle se traduit à la fois par une diminution du temps nécessaire aux opérations de transfert et du risque d'erreurs de transcription des données potentiellement très importantes lors de la saisie de celles-ci sur le clavier.

PRESENTATION DU HUSKY-HUNTER

Le micro-ordinateur Hunter est un système véritablement portable, totalement autonome. Un interpréteur Basic étendu réside dans la mémoire permanente. Le Hunter a également un puissant système de commandes en simulation disque dans la RAM.

1- Présentation.

De fabrication anglaise, par HUSKY COMPUTERS LTD, le Hunter est le plus avancé parmi les véritables micro-ordinateurs portables. Il est totalement autonome.

Dimensions : 21,6 x 15,6 x 3,2 cm.

Poids : 1,2 Kg maximum.

Clavier.

C'est un clavier classique en QWERTY possédant 58 touches. Quelques unes de ces touches ont une position non standard.

Ecran.

C'est un des plus larges écrans à cristaux liquides (LCD). Il peut représenter jusqu'à 320 caractères sur 8 lignes de 40 caractères chacune.

Il possède un affichage graphique de 240 x 64 points.

Ecran virtuel.

L'affichage de 8 x 40 du LCD représente 1/6 de l'écran virtuel. Celui-ci est de 80 caractères sur 24 lignes.

Communications.

Pour les communications, le Hunter est équipé

.../...

d'une sortie-série standard RS-232/V24 et peut être mis en liaison avec la plupart des ordinateurs, périphériques et autres systèmes. En ce qui concerne le programme LPC.BAS, la sortie-série a servi pour sortir les listings sur imprimante Epson ou pour transférer les données.

La sortie communique par l'intermédiaire d'une prise mâle de type D de 25 broches.

Alimentation.

L'alimentation du Hunter se fait par des piles standard.

2- Système.

DEMOS-Système.

Le système DEMOS (Disk Emulating Operating System) gère des fichiers et exécute des programmes CP/M.

Il permet une gestion de programmes et du Basic, une sélection d'options comme mise en place des paramètres de communications, transfert de fichiers par la sortie-série etc...

Structure des fichiers et organisation mémoire.(fig.2)

Le Hunter a la possibilité de stocker dans sa mémoire des programmes Basic, des fichiers textes, des fichiers de données.

La mémoire est gérée comme une mémoire à disques.

Les fichiers ASCII peuvent être affichés à l'écran. Les fichiers peuvent être supprimés, renommés ou sauvegardés. Ils peuvent être déplacés et facilement chargés à l'aide du port-série.

La RAM peut être adressée comme si elle était un disque géré par des routines système.

Cette version du Hunter dont je disposais est conçue pour un maximum de 80 Ko de RAM et 48 Ko de ROM.

La mémoire se décompose ainsi :

- ROM : 48 Ko
 - Basic, communications, écran ...
- RAM : 80 Ko
 - Mémoire de travail : 64 Ko.
 - . 48 Ko de mémoire de travail proprement dite
 - . 10 Ko de RAM commune
 - . 6 Ko d'extension de RAM
 - Espace fichier : 16 Ko.
 - . 14 Ko d'espace fichier proprement dit
 - . 2 Ko de "directory"




Fig. 2

L'espace fichier sur la version du Hunter disponible est très réduit. En effet, il n'a pas été possible de stocker la version définitive du programme LPC.BAS sur celui-ci.

Il sera donc nécessaire d'étendre l'espace fichier pour permettre à la fois la sauvegarde du programme et le stockage du fichier des données saisies.

3- Basic.

L'interpréteur Basic étendu réside en mémoire permanente (ROM).

Le Basic du Hunter se rapproche beaucoup du Basic Microsoft.

LE PROGRAMME

1- Caractéristiques.

Ce programme a donc été conçu en langage Basic sur Husky-Hunter. Il est stocké sous le nom de LPC.BAS.

Il permet la création d'un fichier de données obtenues par l'utilisation de la sonde LPC-INRA.

2- Présentation globale.

En début de programme, à partir d'un menu, l'utilisateur oriente l'exécution vers trois grands "sous-programmes".

Premier sous-programme : entrée des mesures.

Deuxième sous programme : édition des mesures.

Troisième sous-programme : transfert de mesures.

Une option "sortie de programme" est possible à partir du menu.

.../...

3- Présentation détaillée.

Dans cette partie du rapport, nous allons prendre chacune des trois grandes options proposées dans le menu et à l'intérieur de chacune d'elles, décortiquer l'analyse pour mieux comprendre le programme.

A. Entrée des mesures.

La création du fichier où sont stockées les données se fait dans ce "sous-programme".

Il s'agit d'un fichier séquentiel (en ASCII) où toutes les données sont stockées sous forme de chaînes de caractères. Nous avons choisi ce mode de stockage car la programmation Basic du transfert des données par chaînes de caractères est facile à mettre en oeuvre.

Nous avons renoncé à stocker les données dans un tableau de travail. En effet, les variables stockées sont nombreuses et de nature très différente. De plus, le stockage des données ne peut pas répondre à un protocole de saisie rigoureux imposé par le remplissage des tableaux.

En outre, l'organisation du fichier séquentiel obtenu est proche de l'organisation du fichier source du programme de traitement sur MINI 6, ce qui réduit les opérations de manipulation d'un tel fichier dans les étapes de transfert.

Description du fichier :

par profil

- numéro du profil
- date
- caractérisation
- numéro de sonde
- numéro d'écartement
- temps de comptage
- nombre d'humidité
- erreur d'humidité
- par profondeur
 - profondeur
 - premier comptage
 - deuxième comptage
 - troisième comptage
- message de fin de profondeurs
- test de mesures d'écartement
- première cote de mesure d'écartement
- deuxième cote de mesure d'écartement
- première mesure d'écartement
- deuxième mesure d'écartement
- test de commentaire
- commentaire

message de fin de profils

.../...

Ce sous-programme "Entrée des mesures" est à structure modulaire.

- Module 1 : Caractéristiques générales.

Le module 1 correspond à l'entrée dans le fichier de données générales concernant l'utilisation de la sonde.

Date du jour, établie à partir de la date résidant sous système.

Numéro de la sonde : critère pour la recherche de l'étalonnage de référence de la sonde.

Variable ne pouvant prendre que 3 valeurs: 1, 2 ou 3.

Ecartement nominal : écartement entre la source et le détecteur. A partir de cette valeur, on affecte un numéro d'écartement (1, 2 ou 3) qui sera enregistré dans le fichier.

Temps de comptage : seuls les temps de comptage de 10.0, 30.0, 100.0 ou 300.0 disponibles pour les mesures peuvent être stockés.

Nombre d'humidité : nombre de répétitions de la mesure de teneur en eau pondérale nécessaire pour le calcul de la masse volumique sèche.

Erreur d'humidité admise : cette valeur constitue une estimation de la variance associée à la mesure de la teneur en eau, mais seulement si le nombre d'humidité est égal à 1.

Dans le cas contraire, c'est à dire si le nombre d'humidité est égal à 2 ou 3, le programme sur MINI 6 calcule une estimation de la variance à partir des données disponibles.

Après l'entrée de ces caractéristiques générales, une demande de validation est faite à l'utilisateur avec modification d'une seule ligne à la fois dans le cas d'une non validation, ceci grâce à l'entrée du numéro de ligne à modifier.

- Module 2 : Cotes de mesures d'écartement.

Les cotes de mesures d'écartement sont les deux cotes auxquelles seront effectuées les mesures d'écartement pour tenir compte des erreurs dues au non parallélisme des deux tubes porteurs.

L'entrée du signe devant la valeur est obligatoire.

Un test est effectué sur les deux valeurs car la première est strictement inférieure à la deuxième.

.../...

Après l'entrée de ces deux valeurs, une validation avec une demande du numéro de ligne à modifier, dans le cas d'une non validation, est faite.

- Module 3 : Préprogrammation des profondeurs.

Cette partie-là du programme permet la création d'un tableau dans lequel se trouveront les profondeurs de scrutation souhaitées par l'utilisateur.

Ce tableau est dimensionné par le nombre de profondeurs rentré par l'utilisateur. Il est utilisé pour un rappel automatique des profondeurs avant l'entrée des comptages. Le nombre maximum autorisé de profondeurs est égal à 20.

Les valeurs enregistrées ne peuvent être supérieures à 50 cm (longueur maximale utile de la sonde).

- Module 4 : Entrée des données concernant un profil.

L'entrée des données concernant un profil se divise en 4 parties :

- . Caractérisation du profil
- . Pour chaque profondeur scrutée, entrée de 3 comptages
- . Mesures d'écartement
- . Commentaire du profil

Avant chaque profil, une demande de modification des caractéristiques générales est faite à l'utilisateur, sauf pour le premier profil. En effet, au cours d'une manipulation, on peut être amené à changer une ou plusieurs des caractéristiques générales, par exemple le numéro de sonde dans le cas d'une défaillance technique de la sonde utilisée.

- . Caractérisation du profil.
Cela correspond à une ligne de commentaire concernant la localisation du profil (traitement expérimental considéré).
- . Entrée des 3 comptages.
Pour chaque profondeur préprogrammée, l'utilisateur va mesurer à partir de la sonde 3 mesures de masse volumique qui seront donc enregistrées sur le fichier.
Si nécessaire, l'utilisateur peut modifier

.../...

la profondeur préprogrammée. La nouvelle valeur doit à la fois être inférieure à 50 cm et supérieure à la dernière profondeur scrutée.

Au pas suivant, la valeur de profondeur préprogrammée correspond à celle qui est immédiatement supérieure.

Après l'entrée des 3 comptages, une validation est demandée avec modification possible des comptages avant l'enregistrement des profondeurs de scrutation et des comptages. L'utilisateur a la possibilité d'arrêter la saisie des mesures. Il passe à l'étape suivante.

. Mesures d'écartement.

Les mesures d'écartement correspondent au nombre de centimètres qui séparent les 2 tubes aux cotes déjà rentrées par l'utilisateur.

. Commentaire.

C'est une chaîne de caractères permettant de commenter le profil.

A la fin de chaque profil, l'utilisateur a la possibilité de continuer ou d'arrêter les mesures. Dans le deuxième cas, on retourne au menu principal.

B. Edition des mesures.

Ce "sous-programme" permet donc d'éditer le fichier créé à l'écran sur le Hunter.

L'utilisateur a le choix entre une lecture complète ou une lecture partielle, d'un profil particulier.

Dans le cas d'une lecture complète, cela consiste donc en une lecture séquentielle du fichier depuis le début jusqu'à la fin, avec un test au niveau du message de fin de profondeurs et un autre au niveau du message de fin de profils. La lecture s'arrête dès l'apparition de la chaîne "fin".

Dans le cas d'une lecture partielle, on demande à l'utilisateur le numéro du profil à éditer, et après recherche du profil demandé, on édite les données concernées. La recherche du profil à lire se fait par un test du numéro du profil et de la chaîne de caractères "numéro".

.../...

C. Transfert des mesures.

Ce "sous-programme" concerne le transfert des données stockées sur Hunter vers un micro-ordinateur Goupil 3-PC.

Sur le Hunter, en ce qui concerne l'émission, le programme est très simple. Il s'agit d'une lecture séquentielle du fichier et un envoi des données lues par la sortie-série jusqu'à ce que le message "fin" soit lu.

Sur le Goupil, le transfert se fait par la fonction "COPY" qui permet la réception de données byte par byte à partir de la sortie-série et le stockage sur disquettes des données reçues.

Seulement, la sortie-série ne tient pas compte du caractère ASCII "retour chariot". Les données sont donc stockées les unes à la suite des autres. Pour faciliter la lecture du fichier sur Goupil, nous avons créé, entre chaque donnée transmise, un caractère spécifique qui sera testé pour faciliter la restructuration du fichier.

Le 'contrôle Z' envoyé à partir du Hunter après les données du fichier va permettre au Goupil de savoir que celui-ci est terminé et qu'il peut stocker les données reçues sur disquette.

Cette solution est provisoire. L'écriture d'un programme de réception en assembleur est envisagé pour régler le problème du transfert.

CONCLUSION

Ce stage, effectué à la Station Science du Sol du Centre de Recherches d'Avignon, a été fructueux à deux points de vue, essentiellement.

Tout d'abord, sur le plan informatique, il m'a permis de travailler pour la première fois sur un portable, ce qui demande une pratique différente des micro-ordinateurs que j'avais l'habitude de manipuler. Au niveau de la programmation, il m'a permis d'établir un programme de saisies de données, avec toutes les contraintes que cela comporte, c'est à dire faciliter et réduire les manipulations de l'utilisateur sur le portable.

Ensuite, sur le plan "vie d'entreprise", j'ai pu, pendant ces deux mois, côtoyer des scientifiques et des ingénieurs, voir comment fonctionne une station de recherches.

Je tiens à dire l'accueil très sympathique que m'a réservé tout le personnel de la Station Science du Sol, ainsi que le personnel des autres stations, qui m'ont permis de travailler dans une ambiance décontractée et sérieuse.

LISTE DES VARIABLES UTILISEES DANS LE PROGRAMME LPC.BAS

Menu.

CHOIX : réponse au menu proposé.

Entrée des mesures.

NOM\$: nom du fichier donné par l'utilisateur.

- Caractéristiques générales.

JOUR\$: date du jour.
NS\$: numéro de la sonde.
RP\$: écartement nominal.
NE\$: numéro d'écartement.
TC\$: temps de comptage.
NH\$: nombre d'humidité.
EH\$: erreur d'humidité admise.
NUM : numéro de la ligne à modifier.

- Cotes de mesures d'écartement.

Z0\$: première cote.
Z1\$: deuxième cote.
NUM : numéro de ligne à modifier.

- Préprogrammation des profondeurs.

NPRD : nombre de profondeurs.
T : tableau.
I : indice du tableau.
L : indice de ligne.
NUM : numéro de profondeurs à modifier.

- Données concernant un profil.

M3 : test sur le 1er profil.
I\$: numéro du profil.

. Caractérisation du profil.
CAR\$: caractérisation.

. Entrée des comptages.

B2 : numéro de la profondeur.
J : indice du tableau.
PR : nouvelle profondeur rentrée.
variable enregistrée sur le fichier.
PC : valeur de la profondeur précédant la
profondeur courante.

.../...

K : indice de boucle.)var. permettant de
P : compteur.)sauter certaines
)profondeurs
C1 : 1er comptage.
C2 : 2ème comptage.
C3 : 3ème comptage.
NUM : numéro de comptage à modifier.
M2\$: message de fin de profondeurs.

. Mesures d'écartement.
E0\$: 1ère mesure d'écartement.
E1\$: 2ème mesure d'écartement.
NUM : numéro d'écartement à modifier.
T1\$: test de mesures d'écartement.
. Commentaire du profil.
C\$: commentaire.
T2\$: test de commentaire.

B1 : compteur comparé à 60 (nombre maximum de
profils).

M1\$: message de fin de profils.

Edition des mesures.

Toutes les variables utilisées dans "Edition des
mesures" sont les mêmes que celles utilisées dans
"Entrée des mesures", sauf :

NUM : numéro du profil à éditer dans le cas
d'une lecture partielle.
P : compteur utilisé dans la recherche du
profil à éditer.
K : numéro de profondeur (non enregistré dans
le fichier).
T\$: variable parcourant le fichier séquentiel.

Transfert des mesures.

NOM\$: nom du fichier à transférer.
T\$: variable utilisée pour la lecture et le
transfert du fichier.

Correction du nom du fichier.


NOM\$: nouveau nom du fichier.
ES : test sur la provenance de l'erreur.


Pose d'attente.

Z\$: entrée au clavier.


.../...


ORGANIGRAMMES








Caracteristiques générales


























PROGRAMMATION

```
10 CLEAR:CLS:MAXFILES=1:SCREEN1:CUROFF:KEYOFF
15 ONPOWERRESUME
20 CHAR0:LOCATE0,0:PRINT"Bonjour"
30 CHAR4:LOCATE9,2:PRINT"I.N.R.A."
40 CHAR0:LOCATE0,45:PRINT"SCIENCE DU SOL(AVIGNON)"
50 LOCATE99,55:PRINT" c copyright marmey st."
60 CIRCLE(107,59),4
70 FORI=1TO2000:NEXTI:SOUND50,125
210 CLS:SCREEN0:CURON
220 LOCATE13,0:PRINT"MENU PRINCIPAL"
230 LOCATE13,1:PRINT"-----"
240 LOCATE0,3:PRINT"1 Entree mesures      2_Edition mesures"
250 LOCATE0,5:PRINT"3_Transfert mesures  4_Sortie programme"
270 LOCATE17,7:INPUTUSING("9",,1,E)"Votre CHOIX --->",CHOIX
275 IF(CHOIX<1)OR(CHOIX>4)THEN277ELSE280
277 LOCATE33,7:PRINT"  ":BEEP:GOTO270
280 ONCHOIXGOTO500,3000,5000,20000
500 '-----ENTREE DES MESURES-----
510 CURON:CLS:KEYOFF:LOCATE11,3:PRINT"ENTREE DES MESURES"
520 M3=0
525 LOCATE11,6:INPUTUSING(A$,,1,E)"Etes-vous sur?(O/N)",RP$
530 IFRP$="n"THEN210
540 IFRP$<>"o"THEN525
550 CLS:LOCATE0,2:PRINT"Quel est le nom du fichier a creer?"
555 CURON:LOCATE2,4:INPUTUSING("",1,8)" ",NOM$
560 OPENNOM$FOROUTPUTAS1
565 CLOSE#1
567 OPENNOM$FORAPPENDAS1
570 CLS:JOUR$=DATE$
580 JOUR$=MID$(JOUR$,4,3)+MID$(JOUR$,1,3)+MID$(JOUR$,7,4)
585 '-----CARACT. GEN.-----
590 LOCATE5,0:PRINT"CARACTERISTIQUES GENERALES"
600 NUM=0
610 LOCATE0,2:INPUTUSING("9",1,1,E)"1_ Numero de la sonde.....:      ",NS$
612 IFRNS$<>"1"THEN614ELSE617
614 IFRNS$<>"2"THEN615ELSE617
615 IFRNS$<>"3"THEN616ELSE617
616 LOCATE30,2:PRINT"      ":BEEP:GOTO610
617 IFNUM<>0THEN1010
725 LOCATE0,3:INPUTUSING("99.9",4,4,E)"2_ Ecartement nominal.....:      ",RP$
730 IFRP$="15.0"THENNE$="1":GOTO765
740 IFRP$="20.0"THENNE$="2":GOTO765
750 IFRP$="30.0"THENNE$="3":GOTO765
760 LOCATE30,3:PRINT"      ":BEEP:GOTO725
765 IFNUM<>0THEN1010
870 LOCATE0,4:INPUTUSING("D99.9",5,5,E)"3_ Temps de comptage.....:      ",TC$
871 IFTC$<>" 10.0"THEN872ELSE877
872 IFTC$<>" 30.0"THEN873ELSE877
873 IFTC$<>"100.0"THEN874ELSE877
874 IFTC$<>"300.0"THEN875ELSE877
```

```
875 LOCATE29,4:PRINT"           ":BEEP:GOTO870
877 IFNUM<>0THEN1010
981 LOCATE0,5:INPUTUSING("9",,1,E)"4_ Nombre d'humidites.....:           ",NH$
982 IFNH$<>"1"THEN984ELSE1010
984 IFNH$<>"2"THEN986ELSE1010
986 IFNH$<>"3"THEN988ELSE1010
988 LOCATE30,5:PRINT"           ":BEEP:GOTO981
1010 RP$=""
1015 LOCATE0,7:INPUTUSING(A$,,1,E)"Validation (O/N) ",RP$
1020 IFRP$="o"THEN1060
1030 IFRP$<>"n"THEN1010
1040 LOCATE20,7:INPUTUSING(A$,,1,E)"Numero de ligne: ",NUM
1045 IFNUM<10RNUM>4THEN1040
1050 ONNUMGOTO610,725,870,981
1060 IFNH$<>"1"THEN1110
1070 '-----ERR. HUM. ADMISE-----
1080 CLS:LOCATE5,0:PRINT"PRECISION MESURE HUMIDITE"
1085 LOCATE0,3:INPUTUSING("9.999",5,5,E)"5- Erreur humidite admise..: ",EH$
1090 LOCATE0,7:INPUTUSING(A$,,1,E)"Validation (O/N) ",RP$
1095 IFRP$="o"THEN1120
1100 IFRP$<>"n"THEN1090ELSE1085
1110 EH$="0.000"
1120 IFM3=1THEN1380
1125 '-----COTES MES. D'ECART.-----
1130 NUM=0:CLS:LOCATE5,0:PRINT"COTES MESURE ECARTEMENT"
1135 LOCATE0,2:INPUTUSING("ND9.9",5,5,E)"1- Premiere cote (cm)...: ",Z0
1140 IFNUM<>0THEN1155
1145 LOCATE0,3:INPUTUSING("ND9.9",5,5,E)"2- Deuxieme cote (cm)...: ",Z1
1150 IFZ1<=Z0THENBEEP:GOTO1145
1155 LOCATE0,7:INPUTUSING(A$,,1,E)"Validation (O/N) ",RP$
1160 IFRP$="o"THEN1195
1165 IFRP$<>"n"THEN1155
1170 LOCATE20,7:INPUTUSING(A$,,1,E)"Numero de ligne : ",NUM
1173 IFNUM<10RNUM>4THEN1170
1175 ONNUMGOTO1135,1145
1180 GOTO1155
1190 '-----PREPROG. DES PROF.-----
1195 NUM=0:I=0:L=1:CLS:LOCATE2,0:PRINT"PREPROGRAMMATION DES PROFONDEURS"
1200 LOCATE0,2:INPUT"Nombre de profondeurs ",NPRD
1205 IFNPRD>20THEN1200
1210 LOCATE0,7:INPUTUSING(A$,,1,E)"Validation(O/N) ",RP$
1215 IFRP$="n"THEN1200
1220 IFRP$<>"o"THEN1210
1225 DIMT(NPRD):LOCATE0,2:PRINT"           "
1230 LOCATE0,7:PRINT"           "
1235 I=I+1:IFI>NPRDTHEN1250
1240 L=L+1
1245 IF(I/5)<>INT(I/5)THEN1295
1250 LOCATE0,7:INPUTUSING(A$,,1,E)"Validation?(O/N) ",RP$
1255 IFRP$="o"THEN1287
1260 IFRP$<>"n"THEN1250
1265 LOCATE20,7:INPUTUSING(A$,,1,E)"Numero de prof.: ",NUM
```

```
1270 IFNUM<1ORNUM>(I-1)THEN1265
1275 LOCATE22,NUM+1:PRINT" "
1280 LOCATE22,NUM+1:INPUTUSING("D9.9",4,4,E)" ",T(NUM)
1283 IFT(NUM)>50.0QRT(NUM)<=T(NUM-1)THENBEEP:GOTO1280
1285 GOTO1250
1287 IFI>NPRDTHEN1310
1290 CLS:L=2:LOCATE2,0:PRINT"PREPROGRAMMATION DES PROFONDEURS"
1295 LOCATE0,L:PRINT"Profondeur No ";I
1300 LOCATE16,L:INPUTUSING("D9.9",4,4,E)".....: ",T(I)
1302 IF(T(I)>50.0)OR(T(I)<=T(I-1))THEN1300
1305 GOTO1235
1310 I=0:M1$=""
1340 I=I+1
1350 IFI<>1THEN1360ELSE1380
1360 CLS:LOCATE0,3
1365 INPUTUSING(A$,1,E)"Modification des caracteristiques?(O/N)",RP$
1370 IFRP$="o"THENM3=1:GOTO570
1375 IFRP$<>"n"THEN1360
1380 CLS:SOUND50,125:LOCATE3,3:PRINT"Profil No ";I:GOSUB10000
1385 '-----CARACT. DU PROF.-----
1390 CLS:LOCATE28,0:PRINT"Profil No ";I
1395 LOCATE0,3:PRINT"Caracterisation du profil(20car.max)"
1400 LOCATE0,5:INPUTUSING(" ",20)CAR$
1410 LOCATE0,7:INPUTUSING(A$,1,E)"Validation(O/N) ",RP$
1420 IFRP$="n"THENLOCATE0,5:PRINT" ":GOTO1400
1425 IFRP$<>"o"THEN1410
1426 LOCATE0,7:PRINT" "
1427 I$=STR$(I)
1435 I$="numero "+I$
1440 WRITE#1,I$
1450 WRITE#1,JOUR$
1460 WRITE#1,CAR$
1470 WRITE#1,NS$
1480 WRITE#1,NE$
1490 WRITE#1,TC$
1500 WRITE#1,NH$
1505 IFNH$<>"1"THEN1520
1510 WRITE#1,EH$
1520 '-----ENTREE DES COMPTAGES-----
1530 J=1:B2=1:K=0:M2$="":PC=0
1540 CLS:LOCATE28,0:PRINT"Profil No ";I
1555 LOCATE0,1:PRINT"No de profondeur : ";B2
1560 IFJ>NPRDTHEN1660
1610 LOCATE0,3:PRINT"Profondeur= ",T(J)
1620 LOCATE0,7:INPUTUSING(A$,1,E)"Profondeur correcte?(O/N) ",RP$
1625 LOCATE0,7:PRINT" "
1630 IFRP$="o"THENPR=T(J):J=J+1:GOTO1750
1640 IFRP$<>"n"THEN1620
1660 LOCATE0,3:INPUTUSING("D9.9",4,4,E)"Profondeur= ",PR
1670 IF(PR>50)OR(PR<=PC)THENLOCATE12,3:PRINT" ":GOTO1660
```

```
1672 LOCATE0,7: INPUTUSING(A$, ,1,E) "Validation (O/N) ",RP$
1674 IFRP$="n" THEN1660
1676 IFRP$<>"o" THEN1672
1677 LOCATE0,7: PRINT "
1678 IFJ>NPRD THEN1750
1680 IFPR<T(J) THEN1750
1690 P=0
1700 FORK=JTONPRD
1710 IFPR>=T(K) THENP=P+1
1720 NEXTK
1730 J=J+P
1750 NUM=0
1760 LOCATE0,5: INPUTUSING("D999.9",6,6,E) "C1= ",C1$
1765 IFNUM<>0 THEN1790
1770 LOCATE13,5: INPUTUSING("D999.9",6,6,E) "C2= ",C2$
1775 IFNUM<>0 THEN1790
1780 LOCATE26,5: INPUTUSING("D999.9",6,6,E) "C3= ",C3$
1790 LOCATE0,7: INPUTUSING(A$, ,1,E) "Validation? (O/N) ",RP$
1800 IFRP$="o" THEN1820
1810 IFRP$<>"n" THEN1790
1812 LOCATE20,7: INPUTUSING(A$, ,1,E) "No de comptage: ",NUM
1813 IFNUM<10RNUM>3 THEN1812
1815 ONNUMGOTO1760,1770,1780
1820 LOCATE0,7: PRINT "
1830 PR$=STR$(PR)
1850 WRITE#1,PR$
1860 WRITE#1,C1$
1870 WRITE#1,C2$
1880 WRITE#1,C3$
1900 B2=B2+1: PC=PR
1910 IFPC>=50 THEN1950
1920 LOCATE0,7: INPUTUSING(A$, ,1,E) "<s>:stopper <SPACE>:continuer ",M2$
1930 IFASC(M2$)<>115 ANDASC(M2$)<>32 THENLOCATE35,7: PRINT " ":GOTO1920
1940 IFB2>20 THEN1950
1945 IFASC(M2$)=32 THEN1540
1950 CLS:LOCATE0,3: PRINT"FIN DES PROFONDEURS":GOSUB10000
1960 M2$="stop"
1990 WRITE#1,M2$
2000 Z0$=STR$(Z0): Z1$=STR$(Z1)
2003 '-----MESURES D'ECART.-----
2005 CLS:LOCATE28,0: PRINT"Profil No ";I
2010 LOCATE0,1: INPUTUSING(A$, ,1,E) "Mesures d'ecartement? (O/N) ",RP$
2020 IFRP$="n" THEN2065
2022 IFRP$<>"o" THEN2010
2025 T1$="1": NUM=0
2030 LOCATE2,3: PRINT"Z0= ";Z0$:LOCATE15,3: PRINT"Z1= ";Z1$
2035 LOCATE2,4: INPUTUSING("D9.9",4,4,E) "E0= ",E0$
2037 IFNUM<>0 THEN2050
2040 LOCATE15,4: INPUTUSING("D9.9",4,4,E) "E1= ",E1$
2050 LOCATE0,7: INPUTUSING(A$, ,1,E) "Validation? (O/N) ",RP$
```

```
2057 IFRP$="o"THEN2067
2060 IFRP$<>"n"THEN2050
2062 LOCATE20,7: INPUTUSING(A$,,1,E)"No d'ecart. (Oou1): ",NUM:NUM=NUM+1
2063 IFNUM<>1ANDNUM<>2THEN2062
2064 DNNUMGOTO2035,2040
2065 T1$="0"
2067 WRITE#1,T1$
2070 EF$=STRING$(80," ")
2072 WRITE#1,Z0$
2073 WRITE#1,Z1$
2074 WRITE#1,E0$
2075 WRITE#1,E1$
2077 '-----COMM. DU PROF.-----
2078 CLS:LOCATE28,0:PRINT"Profil No ";I
2082 LOCATE0,2: INPUTUSING(A$,,1,E)"Commentaire(80car. max)?(O/N) ",RP$
2085 IFRP$="n"THEN2115
2087 IFRP$<>"o"THEN2082
2090 T2$="1"
2095 LOCATE1,4: INPUTUSING(" ",,80)C$
2100 LOCATE0,7: INPUTUSING(A$,,1,E)"Validation?(O/N) ",RP$
2110 IFRP$="n"THENLOCATE0,4:PRINTEF$:GOTO2095
2113 IFRP$<>"o"THEN2100ELSE2125
2115 T2$="0"
2125 WRITE#1,T2$
2140 IFT2$="0"THEN2170
2150 WRITE#1,C$
2170 CLS:LOCATE0,3: INPUTUSING(A$,,1,E)"Autre profil?(O/N) ",RP$
2180 IFRP$="o"THEN2190
2185 IFRP$<>"n"THEN2170ELSEM1$="fin"
2190 B1=B1+1
2210 IFB1>60THEN2230
2220 IFM1$=""THEN1340
2230 CLS:LOCATE0,3:PRINT"FIN DES PROFILS":GOSUB10000
2240 M1$="fin"
2260 WRITE#1,M1$
2280 CLOSE#1
2290 GOTO210
3000 '-----EDITION DES MESURES-----
3002 CLS:KEYOFF:LOCATE11,3:PRINT"EDITION DES MESURES"
3004 LOCATE11,6: INPUTUSING(A$,,1,E)"Etes-vous sur?(O/N)",RP$
3006 IFRP$="n"THEN210
3008 IFRP$<>"o"THEN3002
3009 NUM=0
3010 ES=0:CLS:LOCATE2,3:PRINT"Quel est le nom du fichier a editer?"
3011 ONERRORGOTO3600
3012 LOCATE2,5: INPUT " ",NOM$
3016 OPENNOM$FORINPUTAS1
3018 CLS:LOCATE2,2:PRINT"Desirez-vous lire"
3019 LOCATE2,3: INPUTUSING(A$,,1,E)"tout le fichier?(O/N) ",RP$
3020 IFRP$="o"THEN3055
```

```
3022 IFRP$<>"n"THEN3018
3024 CLS:LOCATE2,2:PRINT"Quel est le numero"
3025 LOCATE2,3:INPUTUSING("D9",2,2,E)"du profil a lire? ",NUM
3027 IF(NUM<1)OR(NUM>60)THEN3024
3030 P=0
3040 INPUT#1,T$
3045 IFLEFT$(T$,6)="numero"THENP=P+1
3047 IFT$<>"fin"THEN3050
3048 CUROFF:CLS:LOCATE2,2:PRINT"Profil inexistant.":FORM=1TO1000:NEXTM:CUR(
3049 GOTO3524
3050 IFP<>NUMTHEN3040
3055 L=0:CUROFF
3060 CLS:L=L+1
3070 IFL<>1THEN3100
3075 IFNUM<>0THEN3105
3080 INPUT#1,I$
3090 CLS:LOCATE0,3:PRINT"Profil ";I$:GOTO3110
3100 CLS:LOCATE0,3:PRINT"Profil ";T$:GOTO3110
3105 CLS:LOCATE0,3:PRINT"Profil ";T$
3110 LOCATE0,4:PRINT"-----":GOSUB10000
3120 INPUT#1,JOUR$
3125 INPUT#1,CAR$
3130 CLS:LOCATE0,2:PRINT"Caracterisation du profil:"
3140 LOCATE0,4:PRINTCAR$:GOSUB10000
3145 INPUT#1,NS$
3148 INPUT#1,NE$
3151 INPUT#1,TC$
3154 INPUT#1,NH$
3156 IFNH$<>"1"THEN3160
3158 INPUT#1,EC$
3160 CLS:LOCATE5,0:PRINT"CARACTERISTIQUES GENERALES":FORM=1TO600:NEXTM
3170 LOCATE0,1:PRINT"1_Date.....":JOUR$
3180 LOCATE0,2:PRINT"2_Numero de sonde.....":NS$
3190 IFNE$="1"THENRP=15:GOTO3220
3200 IFNE$="2"THENRP=20:GOTO3220
3210 IFNE$="3"THENRP=30
3220 LOCATE0,3:PRINT"3_Ecartement de travail.....":RP
3230 LOCATE0,4:PRINT"4_Temps de comptage.....":TC$
3240 LOCATE0,5:PRINT"5_Nombre d'humidite.....":NH$
3250 IFNH$="1"THENLOCATE0,6:PRINT"6_Erreur d'humidite admise.":EC$
3260 GOSUB10000
3270 K=0
3280 INPUT#1,PR$
3290 CLS:K=K+1
3310 LOCATE0,1:PRINT"No de profondeur: ";K
3332 INPUT#1,C1$
3334 INPUT#1,C2$
3336 INPUT#1,C3$
3340 LOCATE0,3:PRINT"Profondeur= ";PR$
3350 LOCATE0,5:PRINT"C1= ";C1$
```

```
3360 LOCATE13,5:PRINT"C2= ";C2$
3370 LOCATE26,5:PRINT"C3= ";C3$
3380 GOSUB10000
3390 INPUT#1,PR$
3400 IFPR$<>"stop"THEN3290
3410 CLS:LOCATE0,3:PRINT"FIN DES PROFONDEURS":GOSUB10000
3415 INPUT#1,T1$
3420 IFT1$="0"THEN3490
3425 INPUT#1,Z0$
3430 INPUT#1,Z1$
3435 INPUT#1,E0$
3437 INPUT#1,E1$
3440 CLS:LOCATE0,1:PRINT"Cotes d'ecartement:"
3450 LOCATE2,2:PRINT"Z0= ";Z0$:LOCATE12,2:PRINT"Z1= ";Z1$
3460 LOCATE0,4:PRINT"Mesures d'ecartement:"
3470 LOCATE2,5:PRINT"E0= ";E0$:LOCATE12,5:PRINT"E1= ";E1$
3480 GOSUB10000
3490 INPUT#1,T2$
3492 IFT2$="0"THEN3522
3495 INPUT#1,C$
3510 CLS:LOCATE0,2:PRINT"Commentaire:"
3520 LOCATE1,4:PRINTC$:GOSUB10000
3522 IFNUM=0THEN3535
3523 CLS:LOCATE2,2:PRINT"Fin du profil numero ";NUM:FORM=1TO1000:NEXSTM
3524 CURON:CLS:LOCATE2,2:PRINT"Voulez-vous lire"
3526 LOCATE2,3:INPUTUSING(A$,1,E)"un autre profil?(O/N) ",RP$
3528 IFRP$="o"THENCLOSE#1:OPENNOM$FORINPUTAS1:NUM=0:GOTO3024
3530 IFRP$<>"n"THEN3524ELSE3550
3535 INPUT#1,TS$
3540 IFTS$<>"fin"THEN3060
3550 CUROFF:CLS:LOCATE0,3:PRINT"FIN DES PROFILS":GOSUB10000
3560 CLOSE#1
3570 GOTO210
3600 '-----CORR. DU NOM DU FICHIER-----
3610 CUROFF:LOCATE0,7:PRINT"Fichier non cree.Entrez un nouveau nom."
3620 CLOSE#1:KILL"NOM$.DTA":FORM=1TO1000:NEXSTM
3630 LOCATE2,5:PRINT"
3640 LOCATE0,7:PRINT"
3650 CURON:LOCATE2,5:INPUT" ",NOM$
3660 IFES=0THENRESUME3016ELSERESUME5050
5000 '-----TRANSFERT DES MESURES-----
5010 ES=1:CLS:LOCATE11,3:PRINT"TRANSFERT DES MESURES"
5020 LOCATE11,6:INPUTUSING(A$,1,E)"Etes-vous sur?(O/N)",RP$
5030 IFRP$="n"THEN210
5035 IFRP$<>"o"THEN5020
5040 CLS:LOCATE0,2:PRINT"Quel est le nom du fichier a transferer?"
5043 ONERRORGOTO3600
5045 CURON:LOCATE2,5:INPUTUSING(" ",1,8)" ",NOM$
5050 OPENNOM$FORINPUTAS1
5055 CUROFF:CLS:LOCATE2,3:PRINT"Transfert de",NOM$;" en cours."
5060 INPUT#1,T$:LPRINTT$:LPRINTCHR$(64)
```

```
5070 IFT$<>"fin"THEN5060
5080 LPRINTCHR$(26)
5085 CLS:LOCATE2,5:PRINT"Transfert de",NOM$;" termine.":GOSUB10000
5090 CLOSE#1:GOTO210
10000 '-----POSE D'ATTENTE-----
10010 LOCATE27,7:PRINT"presser ENTER":Z$=""
10030 Z$=INKEY$
10040 IFZ$=""THEN10030
10050 IFASC(Z$)<>13THEN10030
10060 CLS:RETURN
20000 '-----SORTIE DU PROGRAMME-----
20010 CLS:LOCATE10,4:PRINT"SORTIE DU PROGRAMME"
20020 LOCATE10,6:INPUTUSING(A$,1,E)"Etes-vous sur?(O/N)",RP$
20030 IFRP$="n"THEN210
20040 IFRP$<>"o"THEN20020
20050 LOCATE20,7:PRINT"Merci,a bientot..."
20060 KEYON:CURON:END
```